
ВВЕДЕНИЕ 
 

Алгебраическое программирование – это программирование, основанное на 
переписывании. Алгебраическое программирование является расширением функционального 
программирования и применяется при решении задач компьютерной алгебры (таких как 
проблема слов в конечно определенных алгебрах, алгоритмы пополнения Кнута-Бендикса или 
Бухбергера), а также задач, связанных с операционной семантикой языков программирования 
(исполняемые алгебраические спецификации компонентов программного обеспечения, 
определение операционных семантик языков программирования, разработка интерпретаторов и 
прототипов компонентов программного обеспечения и др.). 

В отличие от традиционного подхода, ориентированного на использование канонических 
систем правил переписывания с «очевидной» стратегией их применения, в АПС возможно 
сочетание любых систем правил переписывания и разнообразных стратегий переписывания. 
Такой подход значительно расширяет возможности техник переписывания, поскольку 
возрастает их гибкость и выразительность. АПС интегрирует четыре основные парадигмы 
программирования таким образом, что основная часть программы может быть написана в виде 
систем переписывания, императивное и функциональное программирование используются для 
определения стратегий, логическая парадигма реализуется на базе переписывания, 
использующего встроенную процедуру унификации.  
 Инсерционное программирование, развиваемое в настоящее время в Институте 
кибернетики им. В.М. Глушкова, продолжает традиции алгебраического подхода в прикладной 
теории алгоритмов, программировании и разработке кибернетических систем, основы которого 
были заложены В.М. Глушковым и его школой. В алгебре алгоритмов Глушкова программа 
рассматривается как алгебраически определенное преобразование множества состояний 
информационной среды (память, база данных, многоуровневые распределенные структуры 
данных и т.п.). В инсерционном программировании этот взгляд обобщается на объекты, 
обладающие поведением. Вместо пассивной среды, такой, например, как память, 
рассматривается активная среда, в которой преобразование информации системой 
взаимодействующих агентов определяет поведение этой среды, наблюдаемое пользователями 
информации, производимой внутри среды. Программа в инсерционном программировании 
рассматривается как агент, обладающий поведением. Его погружение в среду изменяет 
поведение этой среды. Переход от программ как преобразователей состояний к программам как 
преобразователям поведений можно сравнить с переходом от точечных к функциональным 
пространствам в математике. 
 Инсерционное программирование – это программирование на базе модели поведения 
агентов в средах  [1]. В основе модели лежит понятие размеченной транзиционной системы, т.е. 
системы, определенной так же, как и автомат, множеством состояний и множеством переходов 
(пары состояний), размеченных действиями или событиями. Формально понятие 
транзиционной системы совпадает с понятием недетерминированного частично определенного 
автомата, однако, в отличие от теории автоматов отношение эквивалентности транзиционных 
систем более сильное. 
 Если для автоматов эквивалентность состояний определяется совпадением языков в 
алфавите входных сигналов, порождаемых этими состояниями, то эквивалентность состояний 
транзиционных систем определяется эквивалентностью деревьев поведений, размеченных 
действиями системы. Это отношение называется бисимуляционной эквивалентностью. Оно 
было введено в работах P. Милнера и Д. Парка еще в 70-х годах. В настоящее время понятие 
транзиционной системы и бисимуляционной эквивалентности используется в качестве основно-
го стандарта в поведенческой теории взаимодействующих процессов в сочетании с различными 
более слабыми, чем бисимуляция, отношениями эквивалентности на множестве состояний 
систем. 

 Содержательные мотивы инсерционного программирования строятся на представлениях 
о поведении систем (агентов), взаимодействующих со своим окружением (средой). Поэтому 
ближе всего к инсерционному программированию находятся современные декларативные 
языки моделирования типа UML, SDL, MSC, а также языки, основанные на алгебрах 
взаимодействующих процессов (CCS, CSP, и т.п.). Агентное программирование 
концентрируется в большей степени на проблемах интеллектуализации, однако его 
поведенческие аспекты естественным образом компенсируются инсерционным 
программированием. 
 Отличительной чертой инсерционного программирования является формализация 
понятия среды и функции погружения (insertion) агентов в среду. Функция погружения 
определяет композицию среды и агента, результатом которой является новая среда, готовая для 
погружения других агентов. Агенты и среды рассматриваются как объекты разных типов, 
обладающие поведением, представляемым с помощью транзиционных систем, состояния 
которых рассматриваются с точностью до бисимуляционной эквивалентности. 
 В качестве одного из основных примеров среды может служить компьютер, 
рассматриваемый как среда для программных агентов. Погружение программы в компьютер 
изменяет его поведение и превращает в иную среду. Первая программа (или система программ), 
погружаемая в компьютер, обычно представляет собой операционную систему (WINDOWS, 
UNIX и т.л.), которая расширяет возможности взаимодействия компьютера с программными 
агентами и с внешней, пользовательской, средой, поставляющей программные агенты для 
погружения в компьютер. Простая программа старого типа, которая получает данные, 
отрабатывает некоторый алгоритм и успешно завершает свою работу, изменяет поведение 
среды незначительно и лишь на короткое время. И совсем иное – пакеты прикладных программ, 
постоянно готовые к получению запросов (другой тип агентов, погружаемых в вычислительную 
среду) вместе с данными для решения задач из соответствующего класса. Погружение таких 
пакетов или интерактивных программ, взаимодействующих с внешней средой, существенно 
меняет поведение исходной среды. 
 Еще одним впечатляющим примером сложной среды является Интернет. Агенты в этой 
среде не только взаимодействуют друг с другом, но также обладают свойством мобильности и 
имеют возможность перемещаться в пространстве, создаваемом средой. В телекоммуникаци-
онной среде, поддерживающей мобильную связь, перемещение агентов (мобильных телефонов) 
осуществляется физически, а не виртуально, как в Интернете. Агенты и среды могут быть 
устроены иерархически. Любая среда с помещенными в нее агентами может быть закрыта для 
помещения в нее новых агентов извне, и, рассматриваемая как агент, может быть погружена в 
среду верхнего уровня. 
 Важной чертой инсерционного программирования является недетерминированность 
поведения агентов и сред, присущая реальным системам, которые моделируются 
инсерционными программами. Поэтому реализация систем инсерционного программирования, 
в общем случае, требует применения моделирующих программ (симуляторов) вместо 
интерпретаторов, а также постановки целей для получения конкретных результатов. 

В структуре АПС можно выделить три способа структуризации используемых 
элементов: 

 
– основные типы системных объектов; 
– базовые вычислительные механизмы; 
– допустимые языковые конструкции, объединенные в семейство языков АПЛАН. 
 
Система алгебраического программирования организована таким образом, что 

пользователь имеет одновременный доступ ко всем уровням программирования, начиная с 
языка конкретной предметной области и заканчивая уровнем расширения языка си, которое 
называется L2C. Таким образом достигается эффективное распределение сложности 
разрабатываемой программной системы по различным уровням. Поддерживается 

эволюционность процесса программирования, начиная с выполняемой алгебраической 
спецификации через оптимизирующие преобразования до эффективной программы на языке си. 
Самый верхний и специализированный уровень образуют конструкции, связанные с некоторой 
конкретной предметной областью. Они имеют локальный характер, но также активно 
используются в других предметных областях. Это связано с тем, что в базовом АПЛАНе нет 
типов данных. Строго говоря, существует единственный тип данных – алгебраический терм. В 
то же время имеются возможности для использования строгого типизирования, например, при 
программировании в объектно-ориентированном расширении АПЛАНа.  

 
ПРАВИЛА ПЕРЕПИСЫВАНИЯ 
 

Техника программирования в АПС базируется на переписывании термов. Рассмотрим 
простую задачу функционального программирования – вычисление чисел Фибоначчи. Хорошо 
известное рекурсивное определение n-го числа Фибоначчи задается следующей системой 
соотношений: 
 F(0) = 1 , 
 F(1) = 1 , 
 F(n) = F(n - 1) + F(n - 2) 
Данную систему можно рассматривать как (рекурсивное) определение функции F или систему 
правил переписывания, которую можно использовать для вычисления F(n). Чтобы записать на 
языке АПЛАН программу, использующую эту систему, следует начать с предложения 

INCLUDE <std.ap>, 
если файл std.ap находится в том же каталоге, куда Вы собираетесь поместить Вашу 
программу, иначе следует указать путь к файлу std.ap. Это предложение включает некоторые 
стандартные определения, в частности, обеспечивает использование арифметических операций 
“+”, “-”, отношения “=”, символа “,”, а также некоторых других синтаксических понятий, 
определенных в модуле std.ap. Затем надо определить имя системы 
  NAME R; 
и присвоить начальные значения 
 R:=rs(n)( 
 F(0) = 1 , 
 F(1) = 1 , 
 F(n) = F(n - 1) + F(n - 2) 
 ); 

Первая строка присваивания указывает, что значением имени R является система правил 
переписывания (rs), а n – единственная переменная, используемая системой.  

Общий вид определения синтаксиса систем правил переписывания следующий: 
 

 <rewriting system> ::= rs(<list of variables separated by ",">) 
 (<list of rules separated by "," >) 
 <rule> ::= <simple rule> | <conditional rule> 
 <simple rule> ::= <algebraic expression> = <algebraic expression> 
 <conditional rule> ::= <condition> -> <simple rule> 
 <variable> ::= <identifier> 
 

Систему R можно применить, например, к выражению Т=F(10), и данное выражение 
будет преобразовано таким образом: 

F(10)=F(10-1)+F(10-2)=F(9)+F(8). 
В АПС данное преобразование выполняется за один шаг, так как арифметические операции 
являются интерпретируемыми и выполняются по мере возможности на каждом шаге 
переписывания. Следующий шаг переписывания может быть выполнен одним из двух способов 
в зависимости от того, к которому вхождению выражения F(n) применяется третье правило 
системы. Рассмотрим случай, когда выбирается первое вхождение, то есть F(9). Тогда имеем 
новое выражение: 

T = (F(8)+F(7))+F(8) 

Продолжая подобным образом, то есть выбирая всякий раз самое левое (самое внутреннее) 
вхождение, получим: 

T = ((F(7)+F(6))+F(7))+F(8) = 
= (((F(6)+F(5))+F(6))+F(7))+F(8) = 
.................................. 
= (...((F(1)+F(0))+F(1))+....)+F(8) 

Теперь применимы как третье, так и второе правила. Будем действовать по принципу, 
который используется во всех основных стратегиях переписывания в АПС. Он состоит в том, 
что правило, которое фактически применяется, – это первое из правил, применимых на текущем 
этапе вычислений. Таким образом, последующие два шага преобразований приведут к таким 
выражениям: 
 T = (...((1+F(0))+F(1))+....)+F(8) = 
 = (...((1+1)+F(1))+...)+F(8) 
 

В данной ситуации имеется еще одна возможность выбора: вычислить выражение 1+1 
прежде, чем преобразовать F(1). Применение арифметических операций к постоянным 
величинам также можно рассматривать как переписывание с помощью неявно заданных правил 
переписывания (таблиц сложения и умножения). Поэтому в соответствии с левосторонней 
стратегией в первую очередь следует выполнить сложение. Следующий шаг переписывания: 
 T = (...(2+F(1))+...)+F(8) = 
 = (...((2+1)+F(2))+...)+F(8) = 
  = (...(3+F(2))+...)+F(8) = ...= 55+34 = 89 
Для данной системы переписывания тот факт, что при переписывании выбирается лишь самое 
левое вхождение, не существен. Чтобы получить требуемый результат с помощью системы 
правил R, достаточно использовать любую стратегию переписывания (алгоритм переписы-
вания), удовлетворяющую следующим условиям. 
 

1. На каждом шаге переписывания применяется одно из правил системы либо выполня-
ется арифметическая операция. 

2. Выбор правила осуществляется согласно последовательности, в которой правила 
записаны. 

3. Переписывание продолжается до тех пор, пока это возможно, то есть пока существуют 
вхождения, к которым применимы правила системы, либо возможны выполнения 
арифметических операций над числами. 

 
Стратегия, удовлетворяющая приведенным условиям, называется финальной. АПС 

позволяет использовать встроенные стратегии переписывания, а также писать собственные 
стратегии, которые наиболее подходят для решаемой задачи. Вызов встроенной стратегии 
является вызовом внутренней процедуры. Как правило, он имеет вид s(T,R), где s – имя 
стратегии, T – имя алгебраической структуры данных, к которой применяется система, а R – 
имя системы правил переписывания.  

Имеются две встроенные финальные стратегии переписывания: applytb (пошаговое 
применение сверху вниз) и applybt (пошаговое применение снизу вверх). Каждую из них можно 
использовать для вычисления n-го числа Фибоначчи. Следует лишь задать определение имени 
T, присвоить этому имени начальное значение, а также сформировать задание, соответственно 
которому будет вызвана стратегия и результат будет выведен на экран. Соответствующий текст 
имеет следующий вид: 

 
 INCLUDE <std.ap> 
 NAME T; 
 T:=F(10); 
 task:=applytb(T,R),prn(T); 

Здесь “task” – стандартное имя системы. Оно определено в файле std.ар. Оператор prn(T) 
выводит значение Т на экран. Для выполнения задания в среде UNIX нужно воспользоваться 
командным файлом dotask.sh следующего вида: 
 
LD_LIBRARY_PATH=$HOME/aps-2004/lib:$LD_LIBRARY_PATH 
export LD_LIBRARY_PATH 
 
if [ ! -f can_tbl.tbl ]; then 
 ln -s $HOME/aps-2004/bin/can_tbl.tbl can_tbl.tbl  
fi 
 
if [ ! -f proc_tbl.tbl ]; then 
 ln -s $HOME/aps-2004/bin/proc_tbl.tbl proc_tbl.tbl  
fi 
 
if [ ! -f aps ]; then 
 ln -s $HOME/aps-2004/bin/aps aps  
fi 
 
$HOME/aps-2004/bin/aps -i $1 
 
true 
 
который при необходимости требуется отредактировать, указав фактический путь к каталогу 
aps-2004, содержащему системные модули.  
В результате мы получим десятое число Фибоначчи, то есть 89. 
 Обсудим теперь работу стратегий  applytb и applybt. Сперва рассмотрим, как алгебраиче-
ские выражения представляются в виде помеченных деревьев. Вершины такого дерева 
соответствуют подвыражениям данного выражения. Вершины, соответствующие элементарным 
подвыражениям (таким как число или символ), помечены самими этими подвыражениями. 
Каждая вершина, соответствующая неэлементарному подвыражению, помечена главной 
операцией этого подвыражения.  

Вершина, соответствующая подвыражению f(x1,…,xn), где f – n-арная операция, 
соединяется дугами, пронумерованными числами от 1 до n, с вершинами, которые 
соответствуют подвыражениям x1,…,xn. Дерево, соответствующее выражению F(n-1)+F(n-2), 
можно представить одним из двух способов в зависимости от значения символа F. Он может 
быть определен в языке АПЛАН как одноместная операция с помощью оператора 

MARK F(1); 
В этом случае символ F является меткой двух вершин дерева. Если же символ F появляется без 
определения, то он рассматривается как атом, а главной операцией выражения F(х) является 
операция аппликации, которая обозначается как обычная конкатенация двух выражений. 
Первым аргументом этой операции в выражении F(х) является атом F, вторым – выражение х.  
 В основе обеих стратегий (applytb и applybt) лежит левосторонний обход дерева в 
глубину. При обходе каждая вершина посещается дважды: первый раз при движении сверху 
вниз, второй – при движении снизу вверх. Стратегия applytb применяет систему правил к 
данному выражению в соответствии с текущей вершиной, двигаясь к этой вершине сверху. 
Стратегия применяется к данной вершине столько раз, сколько возможно. Стратегия applybt 
работает аналогично, но движение к текущей вершине происходит снизу. Если в ходе полного 
обхода применилось хотя бы одно правило, то обход начинается снова, а стратегия работает до 
тех пор, пока не останется применимых правил. При движении снизу вверх в процессе работы 
стратегии выполняются все упрощения, которые возможны (такие, как вычисление 
арифметических выражений). Поэтому обе стратегии являются финальными, и каждая из них 
может использоваться для преобразования выражения F(n) при любом натуральном n. Систему 
R можно применять и к сложным выражениям, которые содержат вызовы функции F, 
например, к выражению F(F(n)). Однако в этом случае стратегии работают по-разному. 

Стратегия applybt вычислит значение выражения, а стратегия applytb будет выполнять 
последовательность переписываний, которая не завершается: 
 
F(F(10)) = F(F(10)-1)+F(F(10)-2) = 
 (F((F(10)-1)-1)+F((F(10)-1)-2))+F(F(10)-2) = ... 
 
Бесконечного переписывания можно избежать, если использовать правила перписывания с 
условиями (условные правила переписывания). Заметим, что третье правило системы R следует 
применять к выражению F(n) лишь в том случае, когда n есть неотрицательное целое число. 
Соответствующее условие записывается на языке АПЛАН следуюшим образом:  
isint(n)&(n>0). Его можно добавить к третьему правилу. Новая система имеет вид: 
R:=rs(n)( 
 F(0) = 1 , 
 F(1) = 1 , 
 isint(n)&(n>0)->( 
 F(n) = F(n - 1) + F(n - 2) 
 ) 
); 

Теперь, когда третье правило защищено от нежелательных применений, систему R 
можно применять, используя любую финальную стратегию. Если эта система применяется к 
произвольному выражению, содержащему вхождения F, то будут вычислены все подвыражения 
вида F(n), где n – неотрицательное целое число.  

Рассмотрим правила вычисления чисел Фибоначчи с другой точки зрения. Нетрудно 
заметить, что число шагов, выполняемых любой финальной стратегией для вычисления F(n), 
ограничено снизу выражением, содержащим экспоненту. Действительно, после применения 
третьего правила вычисление F(n) сводится к вычислению F(n-1) и F(n-2). Эти вычисления 
будут выполняться независимо. Вычисление F(n-1) сведется к вычислению F(n-2) и F(n-3). 
Таким образом, выражение F(n-2) будет вычисляться дважды, выражение F(n-3) – трижды и т.д. 
Чтобы разобраться в том, как можно было бы улучшить процесс вычислений, обратимся снова 
к вычислению выражения F(10). Будем допускать теперь применение любых алгебраических 
упрощений, а не только выполнение арифметических операций с целыми числами. Имеем: 
T = F(10) = 
 (F(8)+F(7))+F(8) = 2*F(8)+F(7) = 2*(F(7)+F(6))+F(7) = 
 3*F(7)+2*F(6) = 3*(F(6)+F(5))+2*F(6) = 5*F(6)+3*F(5) = ... 
 ... = 89 
Нетрудно найти общую форму записи примененных правил:  

a∗(x+y)+b∗x = (a+b) ∗x+ a∗y  
 
Приведем также специальные случаи этого правила при a=b=1 и при b=1: 

(x+y)+b∗x = (1+b) ∗x+y,  
a∗ (x+y)+ x = (a+1) ∗x+ a∗y 

(x+y)+ x = 2∗x+y 
 
Если добавить эти правила к системе R, переписывание F(n) можно осуществить за число 
шагов, пропорциональное n, при условии применения подходящей стратегии. Стратегии applytb 
и applybt уже непригодны. Каждая из них будет применять третье правило перед тем, как 
применятся новые правила, что приведет к тем же затратам, что и в случае, когда новые 
правила не использовались. Требуемая стратегия должна работать таким образом, чтобы на 
каждом шаге переписывания правило системы переписывания применялось к самому левому 
внешнему вхождению выражения. И среди встроенных стратегий АПС такая стратегия имеется. 
Это стратегия lmt (самый левый-самый внешний, известная стратегия так называемого 
“ленивого” вычисления). Новая система правил такова: 
 R1:=rs(n,a,b,x,y,z)( 
 
 a * (x + y) + b * x = (a + b) * x + a * y, 

 a * (x + y) +     x = (a + 1) * x + a * y, 
 (x + y) + b * x = (1 + b) * x +     y, 
 (x + y) +     x =       2 * x +     y, 
 
 F(0) = 1 , 
 F(1) = 1 , 
 F(n) = F(n-1)+F(n-2) 
 ); 
а новое задание следующее: 
 task:=lmt(T,R1),prn(T); 

Поскольку описание стратегии lmt и некоторых необходимых для ее работы процедур 
содержатся в системных файлах strat.ар и gen_obj.ap, следует включить  в программу соответ-
ствующие предложения INCLUDE < strat.ap> и INCLUDE <gen_obj.ap>. Заметим, что несколь-
ко предложений включения можно заменить одним вида INCLUDE <include_аplan.ap>, содер-
жащем предложения включения системных файлов.  

Недочетом приведенной выше системы является большое число правил. Его можно уст-
ранить, если более внимательно проанализировать общее состояние вычислений. Такое состоя-
ние может быть описано выражением: 

a∗F(n)+b∗F(n-1),  
 
которое получается уже после четвертого применения системы и повторяется через каждые два 
шага. Обозначим функцию, определяемую данным выражением, через f(a,b,n). Нетрудно 
заметить, что при n>0 

F(n) = 1*F(n)+0*F(n-1) = f(1,0,n),  
f(a,b,n) = f(a+b,a,n-1) 

Таким образом, вычисление F можно заменить вычислением f, полагая f(a,b,0)=а и применяя 
такую систему переписывания 
R2:=rs(a,b,n)( 
 f(a,b,0) = а, 
 f(a,b,n) = f(a+b,a,n-1), 
 F(n)     = f(  1,0,n  ) 
); 

Для управления такой системой нужна более простая стратегия, которая применяет 
систему только к самому выражению и не рассматривает его подвыражения. Эта стратегия 
называется appls. Вычисления ускорятся, если символ f определить как трехместную операцию, 
а не атом или имя. Система R2, очевидно, соответствует хорошо известной процедурной 
программе вычисления F(n), но здесь она представлена в алгебраической форме и получена 
исключительно алгебраическими методами без использования процедурного мышления. Это 
основная особенность алгебраического программирования, которая позволяет приводить 
процедурные конструкции к простому виду, сократив их при этом до необходимого минимума, 
а также выражать наиболее существенные свойства алгоритмов в математическом 
(алгебраическом) виде. 

 
ПРОЦЕДУРЫ 
 

Императивные программы представляют собой последовательности операторов, разде-
ленных символами “,” или “;”.  

Императивная программа может быть значением имени и может быть вызвана 
посредством системной команды: 

Синтаксис оператора следующий: 
 

 <statement> ::= <basic statement>| <conditional statement> 
 | <while statement> | <do statement> | <internal call> 
 | <external call> | return | return(<algebraic expression>) 
 | (<program>) 

 <basic statement> ::= <set statement> | <assignment statement> 
 <set statement> ::= <selector> --> <algebraic expression> 
 <assignment statement> ::= <name> := <algebraic expression> 
 <selector> ::= <name> 
 | arg(<selector>,<sequence of expressions separated by ",">) 
 <conditional statement> ::= <condition> -> <statement> 
 | <condition> -> <statement> else <statement> 
 <while statement> ::= while(<condition>, <statement>) 
 <do statement>  ::= do(<name>) 
 <internal call> ::= <internal procedure name> 
 (<actual parameter list>) 
 <internal name> ::= <atom> 
 <external call> ::= <procedure definition or name of procedure 
 definition>(<actual parameter list>) 
  <actual parameter> ::= <algebraic expression> 
 
Синтаксис определения процедуры: 
 
 <procedure definition> ::= proc(<formal parameters list>) 
 <local names> <statement> 
 <local names> ::= loc(<local names list>) | <empty> 
 <formal parameter> ::= <identifier> 
 
Пример программы:  
 
INCLUDE <std.ap> 
NAME R; 
R:=rs(a,b,n) ( 
 f(a,b,n) = a, 
 f(a,b,n) = f(a+b,a,n-1), 
 F(n) = f(1,0,n) 
); 
NAME(T); 
T:=F(10); 
task:=( 
 prog_init(); 
 prn T; 
 yes:=1; 
 while(yes, 
 garb(); 
 applr(T,R) 
 ); 
 prn T 
); 
 

Данная программа имеет типичную структуру. Основная ее часть представляет собой 
систему правил переписывания R. Вызов внешней процедуры prog_init() обеспечивает 
выполнение некоторых первоначальных действий. Операторы garb() и applr(T,R) – это вызовы 
внутренних процедур. Они вызывают внутренние процедуры, написанные на языке си (который 
является языком нижнего уровня в АПС). “applr” – это одна из базовых стратегий 
переписывания. Она применяет систему R к корню терма Т один раз. Если применение 
успешно, то имя yes получает значение 1, в противном случае – 0. Таким образом while-цикл 
равносилен appls, но оператор сборки мусора garb() выполняется после каждого применения R. 
Это существенно, так как сборка мусора указывается в АПС явно.  

Следующий пример – определение процедуры для стратегии applytb. 

NAMES applytb,appl_tb_rec; 
applytb:=proc(t,R)loc(Y)( 
 dowhile(Y,Y:=appl_tb_rec(t,R))Y 
); 

 
appl_tb_rec:=proc(t,R)loc(Y,s,i)( 
 appls(t,R); 
 Y:=yes; 
 forall(s=arg(t,i), 
 Y:=appl_tb_rec(s,R)|/Y 

 ); 
 t:=can(t); 
 return Y 
); 

Второе определение процедуры также используется как функция, поскольку возвращает 
значение Y; выражение “forall(s=arg(t,i),P)” обозначает цикл по всем непосредственным подтер-
мам s терма t и может быть записано с использованием оператора while. Использование таких 
выражений для сокращения записи будет обсуждаться ниже. 
 
ФУНКЦИИ 
 
Имеется два способа введения функций: посредством систем правил переписывания и с 
помощью (внешних или внутренних) процедур, которые возвращают значение. Их можно 
вызывать явно или путем использования в алгебраических выражениях имен определений 
процедур систем правил переписывания либо имен внешних процедур.  
 Следующий пример – это программа, которая печатает таблицу чисел Фибоначчи 
(заданное число N строк), останавливаясь после вывода каждых десяти строк (внутренняя 
процедура wait_end()). Отметим, что внутренние процедуры могут использоваться как функции, 
которые возвращают значение 1. 
 
INCLUDE <std.ap> 
NAMES tab,turn_right,prtab; 
 
tab:=rs(x,y,z,u,n,p)( 
 ( (x,y,z), 0) = (F(0)=1), 
 ( (x,y,z), 1) = (F(0)=1,F(1)=1), 
 ( (x,y,z), 2) = turn_right(((z,y),x)), 
 ( (F(n)=x,p=y,z), u) = tab((F(n+1)=x+y,F(n)=x,(z,p=y)), u-1) 
); 
 
turn_right:=rs(x,y,z)( 
 ((x,y),z) = turn_right(x,y,z) 
); 
 
prtab:=rs(x,y,n,m)( 
(n>0)&(n mod 10 == 0)->( 
 (F(n)=m,y) = prn(F(n)=m)&wait_ent()&y 
), 
 (x,y) = prn(x)&y, 
 (x=y) = prn(x=y)&wait_ent() 
); 
NAMES T,N; 
task:=( 
 N:=15; 
 T:=tab((F(2)=2,F(1)=1,F(0)=1),N), 
 appls(T,prtab); 
 wait_ent() 
); 

СТРУКТУРА АПС 
 

Структуры данных 
 

Основным типом данных в системе является алгебра термов (представленных в виде 
деревьев) TΩ(Z), порожденная множеством первичных объектов Z и операциями сигнатуры Ω. 
Данная алгебра рассматривается как абсолютно свободная Ω-алгебра. Ее расширение до 
алгебры бесконечных (но конечно представленных, или рациональных) деревьев обозначим 
T*

Ω(Z). В качестве значений имен такие структуры могут иметь общие части и могут 
использоваться для представления произвольных помеченных графов (термов, представленнях 
в виде графов). Такая возможность реализована на императивном уровне и обычно не 
учитывается на уровне алгебраического программирования. 

 
Системные объекты 
 

Системные объекты бывают трех типов: алгебраические программы (ар-модули), 
алгебраические модули (а-модули) и интерпретаторы.  

Алгебраические программы – это тексты на языке АПЛАН. Каждая программа содержит 
описание сигнатуры Ω и синтаксических правил построения алгебраических выражений (тер-
мов). Она определяет также множества имен Х и атомов А. Эти объекты, а также числа и 
строки составляют множество первичных объектов Z. Упомянутые выше три множества задают 
тип (Ω,Х,А) ар-модуля. Типы ар-модулей частично упорядочены отношением включения 
(символы сигнатуры Ω рассматриваются вместе с их определениями, которые, в частности, 
включают местность каждого символа). Если (Ω,Х,А)Ì(Ω¢,Х¢,А¢), то будем говорить, что ар-
модуль М типа (Ω¢,Х¢,А¢) принадлежит классу С(Ω,Х,А). Два класса назовем совместимыми, 
если они имеют общую нижнюю грань, которая является их общим подклассом. Параметры 
этого подкласса содержат параметры обоих совместимых классов. Алгебраическая программа 
определяет также и начальные значения имен, которыми являются объекты типа TΩ(Z). 
 Алгебраические модули содержат внутренние представления структур данных, 
определенных в ар-модулях.Они создаются посредством системных команд, которые 
обращаются к ар-модулям как к генераторам новых объектов.Алгебраический модуль М, 
порожденный программой Р, наследует ее тип и начальные значения имен. Понятие а-модуля 
является динамическим. Он имеет состояние, которое может со временем меняться. Изменение 
состояние а-модуля происходит в результате выполнения (посредством интерпретации) 
содержащихся в нем процедур. Упорядочение на множестве типов а-модулей, также как и 
понятие классов СА(Ω,Х,А) а-модулей, определяется аналогично соответствующим понятиям 
для ар-модулей. Таким образом, ар-модули играют по отношению к а-модулям ту же роль, что 
и классы по отношению к объектам в объектно-ориентированном программировании. 
 
Состояния 
 

Состояние а-модуля типа (Ω,Х,А) слагается из двух компонентов. Первый называется 
состоянием памяти и является отображением σ: Х®Т*

Ω(Z). Второй компонент – это отношение 
эквивалентности на множестве вершин Ω-деревьев, являющихся значениями имен. Две 
вершины могут быть отождествлены согласно данному отношению эквивалентности, если они 
помечены одинаковыми операциями и подтермы, порождаемые этими вершинами, изоморфны. 
Если какие-либо две заданные вершины отождествлены, то соответствующие дочерние 
вершины (аргументы операции, помечающей заданные вершины) также отождествляются. 

Системные интерпретаторы 
 
 Системные интерпретаторы – это программы, предназначенные для интерпретации 
процедур, написанных на языке АПЛАН. Они разрабатываются на языке си на основе 
библиотек функций и структур данных для работы с внутренним представлением системных 
структур данных. Соответствующее расширение языка си называется L2C. Каждый 
интерпретатор связан с некоторым типом (Ω,Х,А), определяющим классы СI(Ω,Х,А), к которым 
принадлежит интерпретатор, так же,как и в случае модулей. Тип определяет то, какие 
алгебраические модули могут выполняться с помощью данного интерпретатора: это модули, 
принадлежащие классу, совместимому с классом СА(Ω,Х,А.  
 Каждый интерпретатор определяет операционную семантику языка АПЛАН для данного 
класса а-модулей и обеспечивает эффективную реализацию процедур, функций и статегий 
переписывания для систем, находящихся в данном модуле. 
 
АПЛАН 
Порождение ар-модулей 
 

Синтаксис, используемый для создания ар-модулей, – это синтаксис алгебраических струк-
тур данных. Специальные структуры данных, такие как правила переписывания, процедуры, 
объекты, подчиняются единым синтаксическим правилам. Алгебраическая программа – это 
последовательность предложений. Предложения бывают следующих типов:  
· описания имен, 
· описания отметок, 
· присваивания начальных значений, 
· включения, 
· комментарии. 
 
<name description> ::= NAMES <sequence  of  names separated by "," >; 
 <name> ::= <identifier> 
 
 <mark description> ::= MARK  <sequence  of  mark  descriptions 
 elements separated by "," >; 
 <mark description element> ::=<mark symbol>(<arity>) 
 | <mark symbol>(2, <priority>, "<infix notation>") 
 <mark symbol> ::= <identifier> 
 <arity> ::= <positive integer> | UNDEF 
 <priority> ::= <positive integer> 
 <infix notation> ::= <sequence of signs> 
 
 <initial assignment> ::= <name> := <algebraic expression>; 
 <algebraic expression>::=<primary expression>|<prefix expression> 
 |<application>|<infix expression> 
 <primary expression>::=<integer or rational number> | <string> 
 | <empty object> | <name> | <atom> | VAL <name> 
 |(<algebraic expression>) 
 <empty object>::= () 
 <application>::=<algebraic  expression> <algebraic  expression> 
 <prefix expression>::=<mark   symbol>(<sequence   of   algebraic 
 expressions separated by ",") 
 <infix expression>::=<algebraic  expression><infix notation> 
 <algebraic expression> 
 

В выражении вида ω(х1,…,хn), представленном в префиксной форме, где ω – это символ 
отметки, количество аргументов должно быть равным местности отметки, если местность – 
целое число, и может принимать произвольное значение, если местность имеет значение 

UNDEF. Приоритет выражения вида хωy, представленного в инфиксной форме, определяется 
приоритетом ω. Выражение х должно быть либо первичным, либо аппликацией, либо 
инфиксным выражением, приоритет которого больше приоритета ω; если y – инфиксное 
выражение, то его приоритет должен быть не меньшим, чем приоритет ω.  
 <inclusion> ::=INCLUDE <file name inserted into "<>"> 
 | INCLUDE "<file name >" 

Комментарии выделяются скобками /*   */. Строки – это последовательности символов, 
заключенные в кавычки “ ”. 
Предложение включения INCLUDE x означает, что на место данного предложения следует 
поместить текст модуля х.  

При обработке описания имен новые имена, содержащиеся в описании, добавляются к 
множеству имен. Описания отметок расширяют сигнатуру Ω. Отметки могут использоваться не 
только как алгебраические операции, но и в качестве: функциональных и предикатных 
символов, имен типов, конструкторов структур данных и т.п. По этой причине вместо операции 
или функции используется просто отметка терма. Если при описании отметки использована 
инфиксная запись, то отметка может использоваться для представления выражений в 
инфиксной форме. В этом случае приоритет позволяет обойтись без скобок. Если значение 
местности есть UNDEF, отметка может использоваться с произвольной положительной 
местностью (такую отметку можно считать связанной с бесконечным семейством операций). 
Имеются две предопределенные отметки, описывать которые необязательно. Это бинарная 
аппликация в инфиксной форме записи без обозначения знака операции и отметка 
ARRAY(UNDEF), используемая для построения массивов. В системе аппликация всегда имеет 
наивысший приоритет. Атомы – это идентификаторы, встречающиеся в программе и не 
описанные как имена, отметки или знаки операций в инфиксной форме записи. 

Внутренним представлением алгебраических выражений являются Ω-деревья, способ 
построения которых очевиден. В результате обработки описания имени каждое имя получает 
начальное значение – пустой объект (это единственный объект, существующий изначально, то 
есть имеющийся до присваивания начальных значений). При присваивании начального 
значения вида x:=y значением х становится терм, представленный алгебраическим выражением 
y. При создании такого объекта значения вместо имен не подставляются, кроме случая, когда 
имя z следует после символа VAL. В этом случае произойдет обращение к z, а не к VAL z. 
Использование этого средства дает возможность отождествлять вершины внутреннего 
представления деревьев. Если VAL z=arg(y,p), то после такого присваивания появится 
эквивалентность arg(x,p)=arg(z,p)(ε). 
 При переопределении имени или отметки предыдущее определение удаляется. То же 
происходит и с начальными значениями. Таким образом, невозможно создание объектов, 
содержащих контуры. Действительно, после присваивания начального значения x:=… VAL x… 
все вхождения VAL x должны быть заменены пустым объектом, даже если х было уже 
присвоено значение. 
 
Императивные программы 

 
Синтаксис программ и внешних процедур описан выше. 

В целях более точного описания процесса вычислений, а также семантики базовых 
предложений и передачи параметров, в настоящем разделе вводится формальная модель 
представления состояний модуля. 

Представление состояний. Рассмотрим состояние (σ,ε) модуля типа (Ω,X,A). Согласно 
данному выше определению, σ есть отображение Х в Т*

Ω(Z), а ε – отношение эквивалентности 
на множестве вершин рациональных деревьев из Т*

Ω(Z). Каждая вершина единственным 
образом представляется в виде пары (х,р), называемой Х-вхождением, где хÎХ, р=(i1,…,im) – 
последовательность положительных целых чисел, представляющих путь в дереве (или 
вхождение в терм) σ(x). Существуют очевидные ограничения на составляющие этого пути, 

налагаемые степенями промежуточных вершин. В языке АПЛАН эта вершина является корнем 
подтерма, представленного выражением вида arg(t,(i1,…,im)), где arg – это стандартная бинарная 
операция из Ω, t=σ(x).  

Пусть U={u1,…,um}– множество символов (алфавит) равномощное множеству классов 
эквивалентности отношения ε. Символы U будут обозначать соответствующие классы. 
Выражение вида (x,p)Îu обозначает, таким образом, что (x,p) принадлежит классу, который 
соответствует u. В то же время будем рассматривать символы U как вершины графа, который 
представляет структуры данных, содержащиеся в даннoм модуле в текущем состоянии. 

Пусть (x,p)Îu, σ(x)=t, arg(t,p)=ω(t1,…,tn). Тогда если (x,(p,i))Îvi, i=1,…,n, то будем писать 
u®ω(v1,…,vn) и назовем данное выражение декомпозицией вершины u. Если arg(t,p)ÎZ, то 
декомпозицией вершины u является u®arg(t,p). Декомпозиция класса u не зависит от 
представления (x,p) этого класса и единственным образом определяется классом. Если (x,())Îu, 
то x®u назовем декомпозицией имени х. Множество декомпозиций вершин и имен называется 
вершинным представлением состояния (σ,ε). Нетрудно показать, что вершинное представление 
единственным образом определяет соответствующее состояние и для каждого состояния его 
вершинное представление определено с точностью до переименования вершин. 

Целесообразно расширить понятие представления, считая, что правые части 
декомпозиции являются произвольными конечными термами над TΩ(ZÈU). Используя такое 
расширение, можно удалить некоторые лишние вершины. Вершину u назовем лишней, если она 
встречается в правых частях декомпозиции не более одного раза. Лишняя вершина u может 
быть удалена путем замены ее единственного вхождения правой частью ее декомпозиции. 
Представление, не содержащее лишних вершин, называется минимальным в отличие от 
представления, определенного выше, которое назовем максимальным. 

Вычисление значений. Для алгебраического выражения t можно вычислить два вида 
значений. Значение первого вида обозначается val(t) и принадлежит множеству TΩ(ZÈU). Оно 
выражается с помощью минимального представления текущего состояния. Значение второго 
вида обозначается Val(t) и принадлежит множеству Т*

Ω(Z). Связь двух видов значений задается 
формулой: 

 
 
где τ¥ есть предел τk, а τ – это подстановка правых частей вершины и декомпозиций имен 
вместо вершин и имен в правых частях всех декомпозиций вершин и имен. Значение второго 
вида не зависит от эквивалентности ε и используется в “инвариантных” рассуждениях об 
алгебраических программах. Первый вид используется для точного определения операционной 
семантики процедурных средств. Функция val(t) осуществляет подстановку значений имен и 
приводит выражение к основной канонической форме, используя интерпретаторы операций 
(функций) ϕω. Формальное определение содержит следующие правила: 

 
 
В формулировке правил фигурирует текущее состояние, которое представлено в виде 
декомпозиции вершины (r). Каждое правило можно использовать лишь в том случае, когда 
неприменимо предыдущее. Выражение isfun(f) истинно, если F=nd(f) есть определение 
процедуры или системы правил переписывания. Если F – это определение процедуры, то эта 
процедура выполняется, при этом х служит списком фактических параметров. Значением 

является то значение, которое возвращает процедура. Если F – это система правил 
переписывания, то она применяется к х со стратегией applr. 
 При выполнении процедур могут возникать побочные эффекты, в результате чего 
возможно изменение текущего состояния модуля. По этой причине порядок вычисления 
значений аргументов операции ω является существенным.  

Базовые предложения. Как случае установки, так и в случае присваивания вычисляется 
значение sÎTΩ(ZÈU) правой части. Рассмотрим предложение установки. Если левая часть – это 
имя х, то его декомпозиция заменяется на х®s. Рассмотрим предложение arg(x,p)®t. Значением 
p должна быть последовательность (i1,…,in) целых положительных чисел. Если (x,i1,…,in-1)Îu и 
u®qÎr, то вместо in-го аргумента q подставляется s. Разумеется, местность q не должна быть 
меньше in.  

Присваивание x:=t функционирует иначе. Сперва, если s – вершина, вместо s берется 
правая часть декомпозиции данной вершины. Если правая часть декомпозиции x не является 
вершиной, то действие присваивания равнозначно действию предложения установки. В 
противном случае, если x®u, u®qÎr, декомпозиция u®q заменяется на u®s.  

Внешние вызовы. Формальные параметры и локальные имена временно добавляются к 
модулю в качестве имен. Формальным параметрам присваиваются значения фактических 
параметров, а затем выполняется тело процедуры, после чего формальные параметры и 
локальные имена удаляются из модуля. Предложение возврата порождает значение, которое 
используется, когда процедура встречается в алгебраическом выражении.  

Внутренние вызовы. Обращение к процедурам реализовано на уровне языка си. 
Количество формальных параметров и то, как их передавать (вычислять значения или нет), 
определяется согласно спецификациям внутренней процедуры.  

Выполнение программ. Выполнение программ осуществляется обычным образом. 
Предложения выполняются одно за другим. Условные предложения и циклы имеют обычный 
смысл. Одной интересной особенностью является использование расширения императивного 
языка АПЛАН. Если интерпретатор программы встречает неизвестное базовое предложение, он 
делает попытку преобразовать его с помощью системы переписывания compile, которая 
является стандартным средством АПС. Пользователь может изменить или расширить систему 
compile. Типичный состав системы такой: 
NAMES _p,_sp,_ptr; 
_p :=nil; 
_sp:=nil; 
MARK case(2); 
 
 
 
 
 
compile:= rs(x,y,z,u,i)( 
 define x = `(defcall `(x)), 
 (arg(arg(x,y),z)-->u) = 
 compile(arg(x,conc(y,z))-->u), 
 (arg(x,y)--> z  ) = `(set(x,y,z)), 
 (    x(i)--> z  ) = `(set(x,i,z)), 
 (   (x.y)--> z  ) = `(set_comp(x,y,z)), 
 (      x --> z.u) =   set_obj_comp `(x-->z.u), 
 (      x --> z  ) = `(setname(x,z)), 
 
 dowhile(x,y) = (x, while(y,x)), 
 dowhile x y  = (x, while(y,x)), 
 for(x,y,z,u) = (x, while(y,(u,z))), 
 loop(x)      = while(1,x), 
 
  forall(x=arg(y,i),z) = 
 for(i-->1,i<= `(ART(y)),i:=i+1, 
 x-->arg(y,i); z 

 ), 
 forallw(x=arg(y,i),u,z) = 
 for(i-->1,(i<= `(ART(y))) & u,i:=i+1, 
 x-->arg(y,i); z 
 
 ), 
 forall(x in list y, z) = 
 (_sp-->(_p,_sp); 
 _p-->y; 
 while( `(is_type( _p,((),()) ) ), 
 x -->arg(_p,1); 
 _p-->arg(_p,2); 
 z 
  ); 
 x-->_p; 
 _p -->arg(_sp,1); 
 _sp-->arg(_sp,2); 
 z 
 ), 
 forall(x in set y, z) = 
 (_sp-->(_p,_sp); 
 _p-->y; 
 while(`(is_type( _p,((),()) ) ), 
 x -->arg(_p,1); 
 _p-->arg(_p,2); 
 z 
 ); 
 _p -->arg(_sp,1); 
 _sp-->arg(_sp,2) 
 ), 
 forallw(x in list y, u, z) = 
  (_sp-->(_p,_sp); 
 _p-->y; 
 while(`(is_type( _p,((),()) )&u ), 
 x -->arg(_p,1); 
 _p-->arg(_p,2); 
 z 
 ); 
 x-->_p; 
 _p -->arg(_sp,1); 
 _sp-->arg(_sp,2); 
 u->z 
 ), 
 forallw(x in set y, u, z) = 
 (_sp-->(_p,_sp); 
 _p-->y; 
 while(`(is_type( _p,((),()) )&u ), 
 x -->arg(_p,1); 
  _p-->arg(_p,2); 
 z 
 ); 
 _p -->arg(_sp,1); 
 _sp-->arg(_sp,2) 
 ), 
 case(x,y->z;u)= starg(x,2,y)->z else case(x,u), 
 case(x,y->z  )= starg(x,2,y)->z, 
  case(x,   z  )=               z, 
 
 branch(y->z;u)= y->z else branch(u), 
 branch(y->z  )= y->z, 
 branch(   z  )=    z 
); 

 




